
' {$STAMP BS2}

' {$PBASIC 2.5}

' --------[Program Description]---------------------------------------------------------

' turnstile demo using IR detectors for safety and tailgate control, and 12VDC Mitsumi

' stepper motor for swing gate operation. demo is designed to be used in conjunction with

' any standard access control system providing a NO output to the input on the PDB. at

' present this is demonstrated using a NO button on the PDB.

'

'

' 2/18/09 - allowed IR to run constant to monitor state of having no one in turnstile area

' by setting the IR detector to active high! this maintains the code behavior without

' major renovations and allows for a quicker run. also removed the debug screen - its

' taking up too much process time and isnt needed for the prototype.

'

'---------------------------------------------------------------------------------------

' --------[I/O Definitions]-------------------------------------------------------------

Speaker                 PIN       0             ' speaker control

ExitLed                 PIN       9             ' IR Led output (exit side)

ExitDetect              PIN       10            ' IR Detector input for CW movement (exit)

EntryLed                PIN       11            ' IR Led output (entry side alarm)

EntryDetect             PIN       12            ' IR Detector input for tailgating (entry)

EntryBtn                PIN       15            ' counter-clockwise button input (entry)

Coils                   VAR       OUTB          ' output to stepper coils

' ---------[Constants]------------------------------------------------------------------

Mitsumi                 CON       12            ' steps / revs by type (only 90 degress by 

design)

Howard                  CON       100           '   "      "    "   "

RevSteps                CON       Mitsumi       ' steps per revolution

NumSteps                CON       4             ' use 4-step sequence

LastStep                CON       NumSteps - 1  ' last step in sequence

IrMod                   CON       38500         ' 38.5 kHz (harmonic)

ModTime                 CON       1             ' 1ms

NoDetect1               CON       0             ' exit detector is active low

NoDetect2               CON       0             ' tailgate detector is active low

TAdj                    CON       $100          ' x 1.0 (time adjustment)

FAdj                    CON       $100          ' x 1.0 (frequency adjustment)

#DEFINE Testing = 0

' ---------[Variables]------------------------------------------------------------------

idx                     VAR       Byte          ' loop counter

stpIdx                  VAR       Nib           ' step pointer

stpDelay                VAR       Byte          ' delay for speed control

object1                 VAR       Bit           ' set to exit IR Detector state

object2                 VAR       Bit           ' set to tailgate IR Detector state

SpkrIdx                 VAR       Word          ' count for speaker on alarm

onTime                  VAR       Word          ' duration for freq out

offTime                 VAR       Word          ' duration for freq out

note1                   VAR       Word          ' first tone for FREQOUT

note2                   VAR       Word          ' second tone for FREQOUT

' ------------[EEPROM Data]-------------------------------------------------------------

'                                 _____

'                                  ABAB

Step1                   DATA      %1100

Step2                   DATA      %0110

Step3                   DATA      %0011

Step4                   DATA      %1001

' ----------[Initialization]------------------------------------------------------------

Setup:

  DIRB = %1111                                  ' make P4..P7 outputs

  stpDelay = 5                                  ' set step delay

' ----------[Program Code]--------------------------------------------------------------

Main:

  DO

    IF EntryBtn = 0 THEN                        ' simulates access control card swipe

      GOSUB EntryGate                           ' from a valid card read and grants access

    ENDIF

    GOSUB Tailgate_Alarm                        ' check for tailgaters after grant

    IF (object2 = NoDetect2) THEN               ' no one is standing in the lane

      GOSUB All_Clear

    ELSE

      PAUSE 3000                                ' wait 3 sec to see if they back out

      GOSUB Alarm2                              ' put system in alarm if they are still there

    ENDIF

    GOSUB Exit_Scan                             ' see if someone is coming out

    IF (object1 = NoDetect1) THEN               ' is someone tailgating the exiting person?

      GOSUB All_Clear                           ' if not, then everything is cool

    ELSE

      GOSUB ExitGate                            ' make sure no one is still standing in the gate 

path

    ENDIF

    PAUSE 10

  LOOP

EntryGate:

  FOR idx = 1 TO RevSteps                       ' 1 rev forward

    GOSUB Step_Fwd

  NEXT

  PAUSE 2000

  FOR idx = 1 TO RevSteps                       ' 1 rev back (return to home position)

    GOSUB Step_Rev

  NEXT

  PAUSE 2000

  RETURN

ExitGate:

  FOR idx = 1 TO RevSteps                       ' 1 rev back

    GOSUB Step_Rev

  NEXT

  PAUSE 2000

  GOSUB Alarm1                                  ' check those tailgaters

  FOR idx = 1 TO RevSteps                       ' 1 rev forward (return to home position)

    GOSUB Step_Fwd

  NEXT

  PAUSE 2000

  RETURN

' ---------[Sub-Routines]--------------------------------------------------------------

' turn stepper clockwise one full step

Step_Fwd:

  stpIdx = stpIdx + 1 // NumSteps                ' point to next step

  GOTO Do_Step

' turn stepper counter-clockwise one full step

Step_Rev:

  stpIdx = stpIdx + LastStep // NumSteps         ' point to previous step

  GOTO Do_Step

' read new step data and output to pins

Do_Step:

  READ (Step1 + stpIdx), Coils                   ' output new coil data

  PAUSE stpDelay                                 ' pause between steps

  RETURN

' scan IR Detector for change to trigger motor

Exit_Scan:

  FREQOUT ExitLed, ModTime, IrMod                ' module IR LED (for exit)

  object1 = ExitDetect                           ' scan detector

  RETURN

' scan tailgate (entry) IR for change to trigger alarm

Tailgate_Alarm:

  FREQOUT EntryLed, ModTime, IrMod               ' module for IR LED (for tailgaters)

  object2 = EntryDetect                          ' scan detector

  RETURN

Alarm1:

  DO

    GOSUB Exit_Scan                              ' give the person a chance to move

    IF (object1 = NoDetect1) THEN                ' good they moved

      RETURN

    ELSE                                         ' nope they are being sneaky

      PAUSE 100

      FOR spkrIdx = 1 TO 4

      onTime = 1000 */ TAdj

      note1 = 1400 */ FAdj

      note2 = 2060 */ FAdj

      FREQOUT Speaker, onTime, note1, note2      ' combine 1400 Hz & 2060 Hz

      onTime = 1000 */ TAdj

      note1 = 2450 */ FAdj

      note2 = 2600 */ FAdj

      FREQOUT Speaker, onTime, note1, note2      ' combine 2450 Hz & 2600 Hz

      NEXT

    ENDIF

  LOOP

Alarm2:

  DO

    GOSUB Tailgate_Alarm                         ' got someone being impatient again?

    IF (object2 = NoDetect2) THEN                ' nope? good!

      RETURN

    ELSE

      PAUSE 100                                  ' give them one more chance to step out...

      FOR spkrIdx = 1 TO 4                       ' they didnt, so embarass em!

      onTime = 1000 */ TAdj

      note1 = 1400 */ FAdj

      note2 = 2060 */ FAdj

      FREQOUT Speaker, onTime, note1, note2      ' combine 1400 Hz & 2060 Hz

      onTime = 1000 */ TAdj

      note1 = 2450 */ FAdj

      note2 = 2600 */ FAdj

      FREQOUT Speaker, onTime, note1, note2      ' combine 2450 Hz & 2600 Hz

      NEXT

    ENDIF

  LOOP

All_Clear:

  PAUSE 1

  RETURN

